
FORM

FORM is a symbolic manipulation toolkit

It supports basic algebraic operations, and can do them very
fast

Memory is no limitation, only disk space (no swapping)

Terms are processed one-by-one

Useful for HEP problems: often we have terabytes of terms!

Obtaining FORM

Download FORM 4.2 from:
https://github.com/vermaseren/form/releases

Tutorial: https://www.nikhef.nl/~form/maindir/

documentation/tutorial/online/online.html

Reference manual:
https://www.nikhef.nl/~form/maindir/

documentation/reference/online/online.html

Compile:

./configure

make -j4

make install

Example program

Save the following as prog1.frm:

1 Symbols a,b;

2 Local F = (a+b)^2;

3 Print;

4 .end

and run

form prog1

Example program

1 Symbols a,b; * define symbols

2 Local F = (a+b)^2; * define expression

3 Print; * print the expression

4 .end; * end the program (and sort)

Time = 0.00 sec Generated terms = 3

F Terms in output = 3

Bytes used = 108

F =

b^2 + 2*a*b + a^2;

Operations on terms

Form can mutate terms with id:

1 Symbols a,b,c;

2 Local F = (a+b)^6;

3 id a^2*b = c; * replacement

4 Print;

5 .end

F =

15*c^2 + 15*b^3*c + b^6 + 20*a*b^2*c

+ 6*a*b^5 + 6*a^3*c + a^6;

FORM behaviour

Form always expands terms

Form processes expressions term by term

This means that only 1 term should fit in memory, the rest
can be read/written to disk

Pro: memory is no limitation

Con: operations on expressions are more difficult

When confused why certain operations don’t exist, imagine that
every expression is too big to fit in memory

The following FORM program will run (try it):

1 Auto Symbols x; * all starting with x is a symbol

2 Local F = (x1+x2+x3+x4+x5+x6)^100;

3 .end

Time = 0.45 sec Generated terms = 100000

F 1 Terms left = 100000

Bytes used = 6106364

Time = 1.00 sec Generated terms = 200000

F 1 Terms left = 200000

Bytes used = 12519468

....

Term flow

Sorting

At the end of a module, terms should be sorted to see if terms
will merge

A module is ended with .sort (or .end)

Where to place the sort is up to the user

Sorting: good vs bad

1 Symbols a,b,c,d;

2 Local F = (a+b+c+1)^6;

3 id a = -c+d+1;

4 id b = -d+1;

5 Print;

6 .end

Generates 924 terms and has 1 in the output...

Sorting: good vs bad

1 Symbols a,b,c,d;

2 Local F = (a+b+c+1)^6;

3 id a = -c+d+1;

4 .sort

5 id b = -d+1;

6 .end

First sort:

Generated terms = 462

Terms in output = 28

Second sort:

Generated terms = 84

Terms in output = 1

Identity statements I

Patterns for id statements can be any terms

Use wildcards var? to match any object of the same type

1 S x,y; * short for Symbol

2 L F = x^2 + y;

3 id x? = 5; * match any symbol with x?

4 Print;

5 .end

yields

F = 30

Identity statements II

Patterns can be more complicated:

1 S x,y,n;

2 L F = x^2 + y;

3 id x?^n? = x^(n + 1);

4 Print;

5 .end

yields

F = y^2 + x^3

Identity statements III

Restrictions can be placed by a set {1,..} or a number range
{>5}

A statement can be repeated with repeat

1 S x,n;

2 L F = x^10;

3 repeat id x^n?{>1} = x^(n-1) + x^(n-2);

4 Print;

5 .end

yields

F = 34 + 55*x;

Functions

Functions for non-commutative functions

CFunctions for commutative functions

1 S a,b,c;

2 CF f; * short for CFunctions

3 Local F = f(1,2,c);

4 id f(1,2,b?) = f(1,2,b?+1);

5 Print;

6 .end

yields:

F = f(1,2,c+1);

Ranged wildcards

A wildcard starting with a ? indicates a range:

1 S x;

2 L F = f(1,2,x,3,4);

3 id f(?a,x,?b) = f(?b,?a);

4 Print;

5 .end

yields

F = f(3,4,1,2);

Applying statements to arguments

id-statements are only applied at ground-level:

1 S x,y;

2 L F = f(x*y);

3 id x = 5; * does not match

4 argument f;

5 id x = 6;

6 endargument;

7 Print;

8 .end

yields

F = f(6*y)

If statements

1 S x,y;

2 L F = f(2) + f(5);

3 if (match(f(x?{>4})));

4 id f(x?) = f(x + 1);

5 else;

6 id f(x?) = f(x - 1);

7 endif;

8 Print;

9 .end

yields

F = f(1) + f(6)

Bracketing I

Powers of variables can be extracted

The terms are not nested for real, but information about
brackets can be used in the next module

1 S x,y,z;

2 L F = x*y + x^2*y + x^2*z + 2;

3 Bracket x; * extract powers of x

4 Print;

5 .end

F = + x * ( y )

+ x^2 * ( z + y )

+ 2;

Bracketing II

Brackets can be indexed in the next module

1 S x,y,z;

2 L F = x*y + x^2*y + x^2*z + 2;

3 Bracket x;

4 .sort

5 L G = F[x^2];

6 Print G; * only print G

7 .end

G = z + y

Bracketing III

Bracketed content can be collected in a function if it fits in
memory

1 S x,y,z;

2 L F = x*y + x^2*y + x^2*z + 2;

3 Bracket x;

4 .sort

5 CF f;

6 Collect f;

7 Print;

8 .end

F = f(z + y)*x^2 + f(y)*x + f(2)

Vectors and indices

Contraction and Einstein summation:

1 Index i1,i2,i3;

2 Vector p1,p2,p3;

3 Local F = p1(i1)*(p2(i1)+p3(i3))*(p1(i2)+p2(i3));

4 Print;

5 .end

yields:

F = p1(i1)*p1(i2)*p3(i3) + p1(i1)*p2.p3

+ p1(i2)*p1.p2 + p2(i3)*p1.p2;

Make sure an index does not appear more than twice in a term!

Traces and gamma matrices

1 S D;

2 Index i1=D,i2=D; * D-dimensional indices

3 Vector p1,p2;

4 Local F1 = g_(1, i1, i1); * gamma matrices

5 Local F2 = g_(1, p1, i2);

6 Local F3 = g_(1, p1, p2);

7 tracen 1; * n-dimensional trace of spin line 1

8 Print;

9 .end

F1 = 4*D;

F2 = 4*p1(i2);

F3 = 4*p1.p2;

Feynman rule application

1 S vhhg, gh, gl; * ghost-gluon vertex, ghost, gluon

2 I i1,i2;

3 V Q,p1,p2,p3,p4;

4 CF vx,prop;

5 L F = vx(Q,p1,p2,i1,vhhg)

6 *vx(-p1,-Q,-p2,i2,vhhg)

7 *prop(p1,i1,i2,gl)*prop(p2,gh);

8

9 id prop(p1?,i1?,i2?,gl) = d_(i1,i2)/p1.p1;

10 id prop(p1?,gh) = 1/p1.p1;

11 id vx(p1?,p2?,p3?,i1?,vhhg) = -i_*vx(p1,p2,p3)*p1(i1);

F = vx(-p1,-Q,-p2)*vx(Q,p1,p2)*Q.p1*p1.p1^-1*p2.p2^-1;

Exercises

Write a program that can differentiate polynomials

Write a program that expands ln(1 − x) up to a certain power
of x (see sum_ in manual)

Substitute the expanded form of x = 1 − ey into the result

Try to make the program faster for higher powers of x

